
Lähtekood

Program valimisfarss;

uses Crt;

type

    _nimi = String[10];

    _ik = String[11];

    _aasta = 0 .. 1990;

    _kuu = 0 .. 12;

    _paev = 0 .. 31;

    _kuupaev = record

             paev: _paev;

             kuu: _kuu;

             aasta: _aasta;

             end;

    _korr = 0 .. 100;

    _ringk = String[13];

    kandidaat = record

              eesn: _nimi;

              peren: _nimi;

              ik: _ik;

              synnik: _kuupaev;

              erak: _nimi;

              haali: longint;

              korr: _korr;

              ringk: _ringk;

              end;

    _fkand = file of kandidaat;

    massiiv = record

            num: integer;

            mass: array[1..100] of kandidaat;

            end;

{ Tervitus programmi käivitamisel }

procedure tere;

begin

     clrscr;

     textcolor(15);

     writeln('Teretulemast Valimisfarss v0.1');

     textcolor(7);

     writeln

end;

{ Hüvastijätt programmi sulgemisel }

procedure headaega;

begin

     textcolor(15);

     writeln('See oli Valimisfarss v0.1');

     textcolor(7);

     writeln

end;

{ Tekstifailist kirjefaili kandmine }

procedure sisend_t;

var

   sisse: String;

   valja: String;

   kand: kandidaat;

   t_f: text;

   k_f: _fkand;

begin

     { Tekst }

     textcolor(15);

     writeln('Andmete lugemine tekstifailist');

     textcolor(7);

     { nimed }

     write('Sisend tekstifaili nimi: ');

     readln(sisse);

     write('Väljund andmefaili nimi: ');

     readln(valja);

     { lugemiseks-kirjutamiseks lahti }

     assign(t_f, sisse);

     reset(t_f);

     assign(k_f, valja);

     rewrite(k_f);

     { loe kuni on mida lugeda }

     while not eof(t_f) do

     begin

          readln(t_f, kand.eesn);

          readln(t_f, kand.peren);

          readln(t_f, kand.ik);

          read(t_f, kand.synnik.paev);

          read(t_f, kand.synnik.kuu);

          readln(t_f, kand.synnik.aasta);

          readln(t_f, kand.erak);

          readln(t_f, kand.haali);

          readln(t_f, kand.korr);

          readln(t_f, kand.ringk);

          write(k_f, kand);

     end;

     { sulge }

     close(t_f);

     close(k_f);

end;

{ Andmepaki lisamine }

procedure sisend_a(var kand: massiiv);

var

   sisse: String;

   k_f: _fkand;

begin

     textcolor(15);

     writeln('Andmete lugemine andmefailist');

     textcolor(7);

     { nimi }

     write('Sisend andmefaili nimi: ');

     readln(sisse);

     { ava }

     assign(k_f, sisse);

     reset(k_f);

     { loe kuni on mida lugeda }

     while not eof(k_f) do

     begin

          inc(kand.num);

          read(k_f, kand.mass[kand.num]);

     end;

     { sulge }

     close(k_f);

end;

{ Tekstifaili salvestamine }

procedure salvesta(var kand: massiiv);

var

   sisse: String;

   i: integer;

   t_f: text;

begin

     textcolor(15);

     writeln('Andmete salvestamine tekstifaili');

     textcolor(7);

     { nimi }

     write('Väljund tekstifaili nimi: ');

     readln(sisse);

     { ava }

     assign(t_f, sisse);

     rewrite(t_f);

     writeln(t_f, 'Nr     Ees-    Perenimi      IK         Sünd      Erakond  Hääli Korr Ringkond');

     for i := 1 to kand.num do

     begin

          write(i:2, ' ');

          write(t_f, kand.mass[i].eesn:10, ' ');

          write(t_f, kand.mass[i].peren:10, ' ');

          write(t_f, kand.mass[i].ik:11, ' ');

          write(t_f, kand.mass[i].synnik.paev:2, '/', kand.mass[i].synnik.kuu:2, '/', kand.mass[i].synnik.aasta:4, ' ');

          write(t_f, kand.mass[i].erak:10, ' ');

          write(t_f, kand.mass[i].haali:5, ' ');

          write(t_f, kand.mass[i].korr:3,'% ');

          writeln(t_f, kand.mass[i].ringk);

     end;

     close(t_f);

end;

{ Andmepaki salvestamine }

procedure salv(var kand: massiiv);

var

   sisse: String;

   i: integer;

   k_f: _fkand;

begin

     textcolor(15);

     writeln('Andmete salvestamine andmefaili');

     textcolor(7);

     { nimi }

     write('Väljund andmefaili nimi: ');

     readln(sisse);

     { ava }

     assign(k_f, sisse);

     rewrite(k_f);

     for i := 1 to kand.num do

          write(k_f, kand.mass[i]);

     close(k_f);

end;

{ Andmepaki kuvamine }

procedure ekraanile(var kand: massiiv);

var

   i: integer;

begin

     textcolor(15);

     writeln('Andmete väljastamine ekraanile');

     textcolor(7);

     writeln('Nr     Ees-    Perenimi      IK         Sünd      Erakond  Hääli Korr Ringkond');

     for i := 1 to kand.num do

     begin

          gotoxy(0, wherey);

          write(i:2, ' ');

          write(kand.mass[i].eesn:10, ' ');

          write(kand.mass[i].peren:10, ' ');

          write(kand.mass[i].ik:11, ' ');

          write(kand.mass[i].synnik.paev:2, '/', kand.mass[i].synnik.kuu:2, '/', kand.mass[i].synnik.aasta:4, ' ');

          write(kand.mass[i].erak:10, ' ');

          write(kand.mass[i].haali:5, ' ');

          write(kand.mass[i].korr:3,'% ');

          writeln(kand.mass[i].ringk);

     end;

end;

{ Kandidaadi lisamine }

procedure lisa(var kand: massiiv);

var

   i: integer;

begin

     textcolor(15);

     writeln('Kandidaadi lisamine');

     textcolor(7);

     inc(kand.num);

     write('Eesnimi: ');

     readln(kand.mass[kand.num].eesn);

     write('Perenimi: ');

     readln(kand.mass[kand.num].peren);

     write('Isikukood: ');

     readln(kand.mass[kand.num].ik);

     write('Sünni päev kuu aasta: ');

     readln(kand.mass[kand.num].synnik.paev, kand.mass[kand.num].synnik.kuu, kand.mass[kand.num].synnik.aasta);

     write('Erakond: ');

     readln(kand.mass[kand.num].erak);

     write('Hääli: ');

     readln(kand.mass[kand.num].haali);

     write('Korruptiivsus: ');

     readln(kand.mass[kand.num].korr);

     write('Valimsringkond: ');

     readln(kand.mass[kand.num].ringk);

end;

{ Andmepaki Häälte järgi sortimine } 

procedure sordi(var kand: massiiv);

var

   kand_b: kandidaat;

   i, j: integer;

begin

     for j := kand.num-1 downto 1 do

     for i := 1 to kand.num-j do

     if kand.mass[i].haali < kand.mass[i+j].haali then

     begin

          kand_b := kand.mass[i];

          kand.mass[i] := kand.mass[i+j];

          kand.mass[i+j] := kand_b;

     end;

end;

{ Päris oting }

procedure otsing(var kand: massiiv);

var

   kand_b: kandidaat;

   i, j: integer;

   valik: char;

   kes: String;

begin

     textcolor(15);

     writeln('Otsing [P]erenimi [I]sikukood [V]alimisringkond');

     writeln('Pingerida [K]andidaatide [E]rakondade');

     textcolor(7);

     write('Valik: ');

     valik := upcase(readkey);

     writeln;

     { edasi läheb programm }

     case valik of

          'P':  begin

                     write('Perenimi: ');

                     readln(kes);

                     j := 0;

                     for i := 1 to kand.num do

                     if kand.mass[i].peren = kes then

                     begin

                          inc(j);

                          kand.mass[j] := kand.mass[i];

                     end;

                     kand.num := j;

                end;

          'V':  begin

                     write('Valimisringkond: ');

                     readln(kes);

                     j := 0;

                     for i := 1 to kand.num do

                     if kand.mass[i].ringk = kes then

                     begin

                          inc(j);

                          kand.mass[j] := kand.mass[i];

                     end;

                     kand.num := j;

                end;

          'I':  begin

                     write('Isikukood: ');

                     readln(kes);

                     j := 0;

                     for i := 1 to kand.num do

                     if kand.mass[i].ik = kes then

                     begin

                          inc(j);

                          kand.mass[j] := kand.mass[i];

                     end;

                     kand.num := j;

                end;

          'K':  begin

                     sordi(kand);

                end;

          'E':  begin

                     for j := 1 to kand.num do

                     begin

                          sordi(kand);

                          kes := kand.mass[j].erak;

                          i := j + 1;

                          while i <= kand.num do

                          if kand.mass[i].erak = kes then

                          begin

                               kand.mass[j].haali := kand.mass[j].haali + kand.mass[i].haali;

                               kand.mass[i] := kand.mass[kand.num];

                               dec(kand.num);

                          end

                          else inc(i);

                     end;

                     sordi(kand);

                end;

     else

         writeln('Et ei tahagi eriti otsida? Oooookeiii...');

     end;

     ekraanile(kand);

     writeln('Kas salvestada tekstifailina(J/E)? ');

     if upcase(readkey) = 'J' then salvesta(kand);

end;

{ Dummy protseduur passiivsuseks }

procedure passiivne(kand: massiiv);

begin

     otsing(kand);

end;

{ Menüü }

procedure menyy;

var

   valik: char;

   kand: massiiv;

   exxit: integer;

   i, x, y: integer;

begin

     { tyhjad andmed, mitte lõpetada }

     kand.num := 0;

     exxit := 1;

     while (exxit = 1) do

     begin

          { menyy }

          textcolor(15);

          writeln('[T]ekstifail andmefailiks | Lugeda [A]ndmefail | [U]us andmepakk');

          writeln('Andmete [S]alvestamine | Te[K]sti salvestamine');

          writeln('Andmete [P]assiivsed Otsingud | Aktiivsed [O]tsingud | [E]kraanile');

          writeln('[H]ääletamine | Kandidaadi [L]isamine');

          writeln('[V]äljuda');

          textcolor(7);

          { valiku teeb kasutaja }

          write('Valik: ');

          valik := upcase(readkey);

          writeln;

          { edasi läheb programm }

          case valik of

               'T':  sisend_t;       { t-fail }

               'A':  sisend_a(kand); { k-fail }

               'U':  kand.num := 0;  { andmed = 0 }

               'S':  salv(kand);     { kirjed }

               'K':  salvesta(kand); { tekst }

               'O':  otsing(kand);   { aktiivsed otsingud }

               'P':  passiivne(kand);{ passiivsed otsingud }

               'E':  ekraanile(kand);{ ekraanile }

               'L':  lisa(kand);     { kandidaadi lisamine }

               'H':  begin           { hääletamine }

                          ekraanile(kand);

                          repeat

                                write('Anda hääl mitmendale? (0 et lõpetada):    ');

                                gotoxy(wherex - 3, wherey);

                                readln(i);

                                if i <> 0 then inc(kand.mass[i].haali);

                          until i = 0;

                     end;

               'V':  exxit := 0;     { välju }

          end;

     end;

end;

begin

     tere;

     menyy;

     headaega;

end.
Lahenduskäik


Alustasin kõige ebavajalikumast ning -õigemast kohast – menüüst. Ta on selline, kuna ühtki teist vähegi mõistlikku lahendust ei tulnud pähe ja mu meelest pole tal ka mingeid puuduseid.


Väljumine sai menüüsüsteemi sisse ehitatud.


Edasi läksin tekstifaili kirjefailks konverteerimise juurde, kuna kirjefaile oli hädasti kõigi teiste töö osade testimiseks vaja. Lahendasin selle ainsal antud olukorras sobilikul viisil lihtsalt tekstifailist kirje välja lugedes ning kirjefaili ümber kirjutades.


Samal põhjusel kirjutasin järgmiseks kirjefaili laadija, mille tegin nõnda, et massiivi salvestatakse kirjed ning samuti ka loendur, kus on kirjas mitu kirjet meil on.


Sarnastel põhjustel (kriitilisemate osade mugavaks testimiseks vaja) kirjutasin seejärel protseduuri, mis kuvab andmed ekraanile. Kuna seda läks niikuinii vaja, lisasin kohe ära ka võimaluse salvestada andmed tekstifaili.


Seejärel kirjutasin nimepõhise kandidaadiotsimise (ning ühtlasi ka valimisringkonna ningi isikukoodipõhise – nad on kõik täpselt sama kood), kuna see on lihtsam kui pingeread. Tegin selle nõnda, et võetakse ajutine uus sekundaarne kirjeloendur, mis näitab et kirjeid pole, andmepangast (mis tuli protseduuri põhipangast sõltumatuna) otsitakse kuni leitakse sobivaid kirjeid ning kirjutatakse panga alguse peale üle, kusjuures suurendatakse loendurit. Kui kõik kirjed läbi vaadatud, kirjutatakse ajutine loendur pärisloenduriks ning andmepakk saadetakse kuvamisprotsessi.


Kirjutasin kandidaadipõhise pingerea, kuna see oli omakorda lihtsam, kui erakonnapõhine. Selleks lihtsalt sortisin kirjed bubblesorti kasutades häälte arvu järgi.


Kirjutasin erakonnapõhise pingerea, kuna õigupoolest midagi muud polnud enam kirjutada. Kõigepealt sortisin kirjed, et erakonda jääks esindama kandidaat, kellel isiklikult oli kõige rohkem hääli. Seejärel tegin tsükli, mis kõnniks, kuni jõuab viimase kirjeni.. Tsükli sees aga lugesin esimese kandidaadi erakonna nime ning teise tsükli abil kandsin kõigi sama erakonna kandidaatide hääled esimesele kandidaadile üle, samal ajal nüüd ilma häälteta kandidaadid pangast välja visates (kirjutades tema peale kõige viimase kandidaadi). Pingerea sortisin uuesti, kuna kõige rohkem hääli kogunu ei pea ilmtingimata olema ka erakonnas, mis kogus kõige rohkem hääli.


Mingil hetkel kirjutasin väikese võimaluse menüüst andmepakk tühjaks teha – et ei peaks programmi käivitama ja sulgema uute andmete… töötlemiseks (

Kuna tundus et pärastpoole oleks vaja funktsiooni ka andmete salvestamiseks, siis tegin selle loomulikul viisil.


Jagasin kaheks otsingu, passiivseks ning aktiivseks (vastavalt siis kas mõjutab andmepakki või mitte). Seda tegin lisades funktsiooni, mis ise ei tee muud kui kutsub välja otsingut, ent samas lõikab läbi ka massiivi sideme peaaenmepakiga.


Lisasin võimaluse kandidaate lisada. Intuitiivsel viisil.


Lisasin võimaluse kandidaatidele hääli anda, kuna see tundus mõistlik asi, mis seal programmis olla võiks. Küsib kandidaadi järjekorranumbrit ning lisab tolle häältele ühe, kuni anda järjekorranumbriks 0 (sellist liiget ei ole olemas).

Ebakohustuslikust


Ei kirjutanud väga palju ebakohustuslikku, kuna põhiliselt usun, et mitte alati ei ole lisafunktsionaalsus kasulik, eriti kui see tuleb intuitiivsuse, lihtsuse või töövõimsuse arvelt… ehkki antud olukorras ilmselt ei oleks lisafunktsionaalsusest halba olnud, ent siiski.


See, mis lisatud sai, oli sama hästi kui kohustuslik, polnud lihtsalt juhendis mainitud, kuna ilma selleta ei oleks rakendus kuigi terviklik olnud. Kui nüüd näpuga järge ajada..


Kandidaadi lisamine programmi sees – ei hakka ju alati uut tekstifaili kirjutama, seda kirjefailiks tegema ja siis juurde laadima, kui kandidaati lisada vaja on. Ei hakanud kandidaadi eemaldamist kirjutama, ehkki kaalusin seda – tõenäoliselt pole sellist funktsiooni iseenesest vaja, kuna ka diskvalifitseeritud kandidaatide andmeid võiks ju säilitada.


Aktiivne/passiivne otsing – see tundus kah säärane asi, ilma milleta oleks programm nagu ainult vaatlemiseks, mitte töötlemiseks.


Hääletamine – polnud kah ära mainitud, aga mis on valimised ilma hääletamisteta?

Analüüs ja õpitu


Taaskord on väga raske kirjutada, et midagi juurde õpitud sai, või üldse, et nagu sel asjal kohutavalt mõtet oleks olnud, kuna see on siiski päris lihtne mõnetunnine projekt – ent kas palju pikemat ja raskemat olekki mõtet teha? Asi, mis mind siiski üllatada suutis, oli pascali ülimalt kahtlane mäluhaldus, kuna ta ei suutnud mõistlikult protseduurides massiive edasi anda, andes Stack errorit. Ent see on rohkem selline tehniline küsimus, mis ei puutu vast nii väga pascalisse, kui keelde vaid pigem minu poolt kasutatud interpretatsiooni sellest.


Tunnis sai juurde õpitud iseenesest pascali süntaksit ja tema iseärasusi erinevate olukordade lahendamisel, mida tõsi küll ei ole teab mis palju, ent asi seegi. Jällegi ei hinda ma seda eriti oluliseks, kuna konkreetselt pascali süntaksit ei lähe tõenäoliselt mitte kellelgi kunagi mitte hetkekski vaja… kui keegi kunagi hetkeks kuskil koolis informaatika algkursust õpetama ei hakka.


Märksa olulisem ning kasulikum oli vast andmebaasidega tutvumine, kuna andmebaasid on infovallas ju ometigi väga olulisel kohal ning saavad väga suure tõenäosusega iga süsteemitehniku igapäevase töö lahutamatuks osaks. Ehkki mul oli ka varem umbkaudne arusaam, kuda nad töötavad ning mis keerulised kooslused seal mingi moel eksisteerivad, materjaliseerusid nad tunnis natuke selgemaks massiks, mille abil on isegi tõenäoliselt võimalik midagi kasulikku teha – maailma, univerumi ja tsivilisatsiooni jaoks.


Üldjoontes usun, et loeng ja harjutustund ning praktika olid vägagi tasemel ning ülimalt arendavad, materjal võeti väga adekvaatselt ning loogilises formuleeringus läbi ning tõenäoliselt ei tekkinud kellelgi suuremaid probleeme, pakkudes samal ajal siiki ka mõningaid väljakutseid, et õppeprotsess huvitavam ning ka arendavam oleks.


Kas ma pean tõesti terve lehe täis kirjutama?


Ülesande keerukus oli mõõdukas – iseenesest täpselt seesama mida me juba varem teinud olime, lihtsalt siitsealt natuke juurde täksitud, lisatud killuke sortimist ning sutsuke suurem mastaap. Tegelikult oli see ikka paganlikult lihtne. Andke andeks.


Hinnates enese, kui ühemehe meeskonna liikme iseseisvat tööd, jäin ma päris rahule. Arvan, et sain võrdlemisi normaalselt hakkama, tegemata ühtki viga ning täites kõik nõudmised, ning lisades veel nipet-näpet.


Kui nüüd taas tunni peale mõelda, siis meenub, kuidas me Exceliga VB abil makro lindistamist proovisime ning kuda sellest midagi välja ei tulnud. Tähendab, excel lindistas tõepoolest mingi makro, kuid kahjuks ei suutnud ta teha midagi, mida poleks olnud lihtsam teha lihtsalt valemeid kirjutades või siis vajaduse korral ise VB’d kirjutades. Seetõttu arvan, et lihtsuses peitub siiski geniaalsus ning tihti teeb lisafunktsionaalsus oma näilise lihtsusega (nuppe peab ju vähem vajutama) asjad masendavalt keerulisemaks ning kokkuvõttes nii programmeerijale kui ka kasutajale kallimaks ning aeganõudvamaks.


Teine asi, mis tunni analüüsimisel meelde tuleb, on need põrgulikud Jacksoni skeemid, mida ükski normaalne inimene tegelikkuses kunagi ei tee, või kui teeb, siis peale koodi kirjutamist, muutes kogu asja täiesti mõttetuks. Võibolla oleks kasulik jagada kodutööd kaheks – Jacksoni skeem ning programmeerimine. Sel moel saaks üliõpilase mugavast vaimust tema oma nõrkuse abil võitu: nimelt ei hakkaks ükski enesest lugupidav tudeng kirjutama koodi, kui homseks on vaja alles skeemi. Nõnda  vahest ilmneks paremini ka asja tegelik mõte ning tema väärtused. Võibolla.


Oletame et see on siis A4.

